
Abstract

The adoption of a common formalism for describing plan-
ning domains fosters far greater reuse of research and al-
lows more direct comparison of systems and approaches,
and therefore supports faster progress in the field. A com-
mon formalism is a compromise between expressive power
(in which development is strongly driven by potential appli-
cations) and the progress of basic research (which encour-
ages development from well-understood foundations). The
role of a common formalism as a communication medium
for exchange demands that it is provided with a clear se-
mantics. This paper describes extensions ofPDDL2.1 (used
in the 3rd International Planning Competition) that support
the modelling of continuous time-dependent effects and il-
lustrates why they can play a critical role in modelling real
domains.

Introduction

This paper describesPDDL+, a significant extension of
PDDL2.1 intended to support the representation of deter-
ministic real time problem domains involving numeric-
valued resources.PDDL2.1 supports the modelling of du-
rative actions and other metric quantities and was used
in the AIPS 2002 planning competition. It comprises a
core drawn from McDermott’sPDDL (McDermott & the
AIPS’98 Planning Competition Committee 1998; McDer-
mott 2000) supplemented with numeric and durative action
extensions (Fox & Long 2002).PDDL2.1 is limited to the
discrete modelling of time. The only time points that can
be identified in a plan are those associated with the start
and end points of actions selected by the planner. The lan-
guage cannot represent exogenous events and is therefore
of rather limited utility for modelling realistic problems in
which activity is constrained by factors outside the plan-
ner’s control.

An important contribution made byPDDL+ is the abil-
ity to model predictable exogenous events. This enables
a planner to reason about the consequences of certain ex-
ecution failures and to plan to avoid, or to exploit, con-
sequences of its actions that will be brought about by the

world. For example, the event of an instrument malfunc-
tioning will occur when the temperature of a satellite drops
below a threshold. A planner can plan to avoid this hap-
pening by ensuring that it maintains its temperature above
this threshold using a charged battery. This may involve ad-
vance planning to ensure the battery is adequately charged
to heat the satellite during its entire period out of sunlight.

PDDL+ is powerful enough to model domains contain-
ing both discrete and continuous behaviours. The key ex-
tension thatPDDL+ provides is the ability to express tem-
poral behaviour in terms of the initiation and termination
of processesthat act on the numeric components of states.
Concurrent processes can interact resulting in continuous
change. Logical state changes are effected by the instanta-
neous initiation and termination activities marking the end
points of active processes.PDDL+ supports the modelling
of situations in whichPDDL2.1 is insufficiently expressive
and which capture important aspects of time-critical con-
current behaviours.

In this paper we discuss the features ofPDDL+ and
demonstrate its use in modelling situations in which there is
continuous change. We compare the operator-centric view
of PDDL+ with some related work in the modelling of ac-
tion and change and explain the advantages of the operator-
centric view for planning.

Modelling Continuous Time
In PDDL+ one of the numerically varying quantities is time.
The objective is to be able to model continuous processes as
these arise in planning problems. An agent must be able to
interact with continuously changing quantities and to plan
with up-to-date information about their values. The model
of time in PDDL+ is therefore real-valued and continuous.

Several researchers have addressed the problem of
modelling time and numeric quantities in a discretized
way (Bacchus & Kabanza 2000; Smith & Weld 1999). The
basic idea is to add a value to an action specification indi-
cating the duration of that action. For example, the action
of a rover driving between two waypoints would have as-
sociated with it a duration of, say, ten minutes and an asso-

Figure 1: Interactions between actions and events. In each figure,A is an action andE is an event, each with indicated
numeric precondition, wherey is a constant andx is increasing under the influence of an active process.

ciated energy consumption. A difficulty immediately arises
in describing the effect of energy consumption on the en-
ergy level of the rover.

Using discretized effects it is necessary to use a step
function to model the energy consumption. If the step func-
tion is applied at the start of the action execution then the
energy level will appear lower than it actually is for the du-
ration of the action, while if it is applied at the end then
the energy level will appear much higher. In either case,
concurrent actions that depend on the energy level will be
forced to consult an inaccurate value leading to possibly
flawed behaviour. The safe way to avoid this problem is
to adopt the conservative requirement that the energy level
is undefined during the execution of the action, forcing ac-
tions that must access its value to be sequenced before or
after the move.

If an action is associated with a duration then the action
will have its effect at the end of that duration without further
intervention by the planner. This prevents the modelling of
effects that occur as the consequence of an action initiating
some process that then needs to be explicitly terminated. In
many realistic situations the environment itself brings about
change as a consequence (perhaps unintended) of the activ-
ities of an executive. For example, sinks will overflow if
the water source into them is left open (Shanahan 1990)
and soup bowls will spill if they are tilted far enough in one
direction without sufficient compensating tilt on the other
side (Gelfond, Lifschitz, & Rabinov 1991).

To enable the correct modelling of these situations we
have supplementedPDDL2.1 with two additional modelling
components:processesandevents. In PDDL+, as inPDDL

and its precedents, non-durative actions have instantaneous
effects that result in a change in the logical state of the do-
main. If change is modelled as deterministic the planner has
perfect knowledge about action outcomes. Of course, this is
a simplification of real world behaviour but it is the classical
assumption upon whichPDDL+ is based. When grounded
(to remove quantifiers and conditional effects), the action
semantics can be supplied by straightforward mappings to
finite transition systems. Actions can also have numeric
pre- and post-conditions which both consult and modify nu-

meric variables.

Like actions, events are modelled as instantaneous state
transition functions which can have numeric pre- and post-
conditions. They are distinguished from actions only by the
fact that the planner cannot select them in the development
of a plan. This enables change in the domain to be modelled
by means of a transition system in which some of the transi-
tions perform mappings between states (actions and events
do this) whilst others model continuous numeric change.
PDDL+ domains can be modelled as hybrid automata, pro-
viding a simple and clean semantic basis for the language.

Actions and events can initiate the execution of processes
that then run over time. A process maintains the logical
aspects of a state whilst modifying numeric aspects of the
state as time passes. Processes have to be terminated, either
by the deliberate action of the executive or by the environ-
ment itself. Eventsare not under the control of the exec-
utive — their role in a plan is to signal the occurrence of
predictable exogenous events. For example, a foreseeable
event can be triggered to happen at a given absolute time.
Events that are the consequences of initiated actions, such
as the event of a sink overflowing, a soup bowl spilling,
an instrument becoming over-exposed to sunlight or of a
rover becoming sun-lit, occur because their preconditions
become satisfied after some period of time in which one or
more processes have been active.

In hybrid automata theory no distinction is made between
actions (under the control of the executive) and events (un-
der the control of the world). In planning, the key difference
between them is that no matter what plan the executive is
executing events will always take priority over actions if
they become enabled in a state which also satisfies the pre-
conditions of an action. Figure 1 shows how conflicts be-
tween actions and events are resolved in the interpretation
of PDDL+ plans.

A crucial detail of the semantics is that it ensures that
when the preconditions of an event are satisfied then that
event must be the next transition executed. This gives rise to
several questions concerning the relationship between ac-
tions and events. Figure 1 depicts three situations in which
there might be considered to be an action/event conflict. In

Figure 2: Discrete models of continuous behaviours fail to adequately account for concurrent interactions.

part (a) we consider the actionA to be ill-defined because
it relies on the executive being able to apply the action at
a precise point in time: the preconditionx = y is true in-
stantaneously and to exploit it would require the capacity to
measure time to an arbitrary degree of accuracy. We argue
that the executive does not have access to this level of pre-
cision, and that only the world itself is able to synchronize
its activity precisely. In part (b)A andE are well-defined
and can each occur. If the plan includesA timed to exe-
cute at the moment whenx = y then this is valid provided
that the order in whichA andE in fact occur does not af-
fect the outcome (ie: both orderings result in the same state
being reached). In part (c) there is no conflict betweenA
andE since they are never applicable at the same instant.
Note that events can have preconditions that are precisely
synchronized with the world because they are not under the
control of an executive — the world has the capacity to re-
spond to situations with arbitrary precision.

Processes and events are distinguished by the fact that
the numeric post-conditions of events cannot be time-
dependent whilst those of a process always are. A process
can be terminated by an action, or by some event that occurs
in the world (such as a pan boiling dry, or a vehicle run-
ning out of fuel). The introduction of processes and events
means that activities resulting in change occuring over time
are modelled, inPDDL+, by a three-part structure consisting
of a point of initiation, a process and a point of termination.
The points of initiation and termination can be the points of
application of actions or events or they can be the points at
which the effects of active processes cause numeric values
to reach critical thresholds. We refer to this as thestart-
process-stopmodel.

PDDL+ supplies additional expressive power over earlier
PDDL variants includingPDDL2.1. Its primary contribu-

tion is the ability to model the occurrence of events that are
produced by the environment instead of as the direct con-
sequences of the executive’s action. This makes it possible
to model many situations that lie outside the capabilities
of less expressive languages. In particular, it is possible to
model domains in which events that the world controls can
be exploited by the planner to achieve a desired effect.

Processes, Events and Durative Actions
The three examples described in this section demonstrate
the need for the three main modelling components of
PDDL+: durative actions with flexible durations, events and
processes. In the following we will refer to actions the dura-
tions of which can be determined by the planner asflexible
durative actions. All of these components support the mod-
elling of continuous change but flexible durative actions can
encapsulate this change when the period of time over which
a quantity is changing, and the point at which it will termi-
nate, can be predicted by the planner and the change itself is
under the control of the planner. Events are required when
changes are triggered that lie outside of the control of the
planner, and which the planner might need to plan to avoid.
Finally, processes are required when an event initiates a pe-
riod of change.

We present the three examples by first motivating them
and then describing fragments ofPDDL+ which success-
fully capture the intended behaviours. We also present parts
of the hybrid automata that model these descriptions, in or-
der to provide some semantic intuitions. A detailed BNF
description ofPDDL+ is available from (Fox & Long 2002).

Modelling Data Downlink in the Satellite Domain
In the satellite observation scheduling domain one of the
important features is that downlink opportunities coincide



with windows of time in which the satellite is in commu-
nication with a ground station. The times at which these
windows will open and close can be computed from other
information, such as the orbit of the satellite and the posi-
tions of the ground stations. Therefore the opening times
can be predicted and a planner must make use of the oppor-
tunity to downlink data if it wants to maximise the amount
of data collected and transmitted (given that it has a finite
amount of storage space on board).

The purpose of this example is to demonstrate why it is
necessary to have the ability to model continuous change.
Suppose that the behaviour of the on-board storage medium
has been abstracted to a level at which reading from and
writing to store can be treated as happening concurrently.
In this situation it is possible for the satellite to make obser-
vations and downlink stored data concurrently if an obser-
vation window overlaps any part of a downlink window.

At first sight it might appear that the storage and down-
link of data in the situation just described can be modelled
using durative actions with step function effects. Figure 2
shows how records and downlinks can be performed con-
currently, both accessing a data store with a maximum ca-
pacity. As can be seen from the figure, the step function
effects make it impossible for the true status of the storage
device to be correctly modelled. The storage capacity of
the device might be exceeded at the point where the step
function is applied. Equally, the data might be downlinked
before it has actually been recorded. The reason for the
difficulty is that the recording and downlinking of data are
really continuously affecting the amount of stored data in a
way that tightly interacts.

The planner can determine the point at which the storage
capacity would be exceeded and plan to halt the recording
process before this consequence occurs. In order to do this
it is necessary to model the recording action as a durative
action the duration of which is determined by the time it
takes to make the desired observation. The action termi-
nates when the invariant condition (that capacity has not
been exceeded) is violated. Similarly, the action that mod-
els downlinking of data has flexible duration determined by
there being data to transmit. These actions both affect the
amount of data stored, in a way determined by the rate of
storage and transmission, and they can be executed concur-
rently. At any point during their execution the amount of
data stored and the amount so far transmitted can be com-
puted from known values. Figure 3 shows how the two du-
rative actions could be modelled. The special literal#t is
used to denote the time period over which the action is ex-
ecuting. Using#t it is possible to access arbitrary time
points within the execution interval.

(:durative-action observe
:parameters (?r - recorder ?i - instrument

?o - observation)
:duration (= ?duration (observationTime ?i ?o))
:condition (and (at start (targetted ?i ?o))

(over all (targetted ?i ?o))
(over all (<= (data ?r) (capacity ?r))))

:effect (increase (data ?r) (* #t (dataRate ?i))))

(:durative-action downlink
:parameters (?r - recorder ?g - groundStation)
:duration (> ?duration 0)
:condition (and (at start (inView ?g))

(over all (inView ?g))
(over all (> (data ?r) 0)))

:effect (and (increase (downlinked)
(* #t (transmissionRate ?g)))

(decrease (data ?r)
(* #t (transmissionRate ?g)))

Figure 3: Durative actions encoding the continuous be-
haviours of recording and downlinking data.

Maintaining Satellite Operating Temperatures

In the recording and downlinking example it is possible to
model the continuous behaviour using durative actions the
durations of which are expressed usingduration inequali-
ties. The actual duration of any action instance can be com-
puted from variables whose values can be predicted in ad-
vance (such as the point at which the storage tape will be
filled, given the capacity of the tape, the amount of data it
holds at the start of the action and the rate at which data can
be stored). However, there are other examples where peri-
ods of continuous change cannot be bounded in advance.

When the orbit of a satellite takes it out of the sun it be-
gins to cool at a continuous rate. During the cooling period
it must use battery power to keep itself warm enough to
prevent instruments from malfunctioning. The temperature
of the satellite changes continuously as a result of the in-
teraction between the cooling and warming processes. If
the battery runs out of charge at some point the satellite
will cool until a point is reached at which the instruments
will malfunction. This critical point is brought about by an
eventthat triggers when the temperature reaches a critical
threshold.

The cooling of the satellite is triggered by an event that
occurs when the satellite is removed from all heat sources.
This event in turn triggers a process, the cooling process,
which continues for as long as the satellite remains in the
cold. While the cooling process is in operation the tem-
perature of the satellite is falling and further events may be
triggered as thresholds are passed.

It is not possible to model processes that are au-
tonomously triggered by events within the durative action
framework. The planner may not know, or care to know,
the extent over which these processes will be active. It may
be able to plan to counter the cooling effects by, for ex-
ample, using battery power. An important reason why the
cooling process cannot be modelled using a durative action
is that it is not under the planner’s control to choose the



(:action heat
:parameters (?s - satellite)
:precondition (and (> (energy ?s) 0) (heaterOff ?s))
:effect (and (heaterOn ?s) (not (heaterOff ?s))))

(:process cooling
:parameters (?s - satellite)
:precondition (inShade ?s)
:effect (decrease (temperature ?s)

(* #t (F(temperature ?s)))))

(:event freeze
:parameters (?s - satellite ?i - instrument)
:precondition (and (functional ?i) (onBoard ?s ?i)

(< (temperature ?s) (safeValue ?i)))
:effect (not (functional ?i)))

(:process warming
:parameters (?s - satellite)
:precondition (and (heaterOn ?s) (> (energy ?s) 0))
:effect (and (increase (temperature ?s) (* #t (heatRate ?s)))

(decrease (energy ?s)
(* #t (heaterConsumption ?s))))

Figure 4: A fragment of a satellite domain encod-
ing the heating and cooling effects. The expression
F(temperature ?s) indicates the cooling effect due to
radiation (and would be written explicitly in a complete en-
coding). There is a corresponding heating process that oc-
curs due to radiation from the sun when the satellite is not
in shade.

cooling effect. The satellite cools because of the physical
nature of the world. The planner can plan to counter the
effect by ensuring that its battery is charged sufficiently to
heat the satellite during its period in the cold, demonstrating
the difference between processes that are triggered byac-
tionsand processes that are triggered byevents. When the
planner has control over how long they will operate, those
triggered by actions can often be encapsulated in flexible
durative actions. However, the planner may not know or
care how long an autonomously triggered process will last
so those triggered by events must be modelled differently.

Figure 4 shows thePDDL+ description of the cooling and
heating example. The#t literal is used to denote the time
period over which a process instance runs. That is, each
separate continuous period over which the preconditions of
the process are satisfied will have a duration whose value
is denoted by#t for that process instance. It can be seen
that the heating action offsets the effects of the cooling pro-
cess, affecting the time taken to reach the critical threshold.
By planning to store enough charge prior to entering the
cold region the planner can avoid the critical threshold be-
ing reached at all.

Modelling a Continuous Recharging Process
The recharging situation is similar in detail to the cooling
and heating example described above. When the robot en-
ters the sunlight the recharging process begins, triggered
by the event of arriving in the sun. The planner need not
care how long the recharging process continues, provided
that there is enough charge available for all of the robot’s
activities to be completed. Driving and digging activities

(:action activate-charger
:parameters (?r - rover)
:precondition (and (in-sun ?r)

(< (charge ?r) (capacity ?r)))
:effect (is-charging ?r))

(:process charging
:parameters (?r - rover)
:precondition (and (<= (charge ?r) (capacity ?r))

(in-sun ?r)
(charging ?r))

:effect (increase (charge ?r) (* #t (charge-rate ?r))))

(:event stop-charging
:parameters (?r - rover)
:precondition (or (= (charge ?r) (capacity ?r))

(not (in-sun ?r)))
:effect (not (charging ?r)))

Figure 5: A fragment of the rover domain showing the
recharging process with its initiating action and concluding
event.

consume charge, resulting in interacting increasing and de-
creasing effects on charge while these activities are taking
place in the sun. The planner can plan to exploit the charg-
ing effect of the sunlight, but it cannot dictate how long the
charging process will last.

The difference between this example and the cooling ex-
ample is that it might seem reasonable for the charging ef-
fect to be modelled using a durative action (since the charg-
ing effect is a desirable one that the planner might choose).
However, in fact, charging when in the sun is not controlled
by the planner but happens to be a positive effect that the
planner can exploit. Furthermore, using a durative action
would necessitate identifying an upper bound on the time
over which charging would take place. A reasonable up-
per bound might be suggested by the charge capacity, but,
in fact, the potential to continue charging is present for as
long as the robot is in the sun (as long as the robot remains
in sunlight charging resumes as soon as stored charge is de-
pleted). Modelling charging as a process is more accurate
than using durative actions because it makes the world re-
sponsible for the length of the charging period and for the
interaction between charging and charge consumption. The
details can be seen in Figure 5.

Plan Metrics

An important extension that we introduced intoPDDL2.1,
and which is also available inPDDL+, is an (optional) field
within the description of problems to enable the specifica-
tion of plan metrics. Plan metrics specify, for the benefit
of the planner, the basis on which a plan will be evaluated
for a particular problem. The same initial and goal states
might yield entirely different optimal plans given different
plan metrics. The following metric evaluates a satellite ob-
servation plan according to the amount of data successfully
downlinked.

: metric maximize (datadownlinked)



In this expression the valuedatadownlinkedis incremented,
by the amount of data transmitted, every time a downlink-
ing operation is completed in the plan.

The ability to encode such metrics makes it possible for
plans to be evaluated according to more realistic notions of
quality than sequential length. Length is a very coarse mea-
sure of quality that does not take into account the cost or
utility of specific actions. An enriched descriptive power
for the evaluation of plans is a crucial extension for the
practical use of planners, since it is almost never the case
that real plans are evaluated solely by the number of actions
they contain.

Semantics of Domain Models
The traditional semantics for planning languages rests on
state transition models, so it is natural to consider tempo-
ral and metric extensions of finite automata as a foundation
for planning language extensions. The theory of hybrid au-
tomata (Henzinger 1996; Gupta, Henziner, & Jagadeesan
1997; Henzinger & Raskin 2000), which has been a fo-
cus of interest in the model-checking community for some
years, provides an ideal formal basis for the development
of a semantics forPDDL+. Our examination of this theory
revealed that the issues we have considered in the devel-
opment ofPDDL+ have been considered and, in many im-
portant cases, resolved, by research done in this field. Our
main contribution has therefore been to develop a language
for succinct encodings of hybrid automata, in a form that is
directed for use in planning. Our hope is that both the for-
mal (semantics and other properties) and practical (model-
checking techniques) results in hybrid automata theory will
be able to be exploited by the planning community in ad-
dressing the problem of planning for discrete-continuous
planning domains.

We have developed the semantics ofPDDL+ by means
of a mapping to hybrid automata, effectively demonstrating
how the succinct encoding of a domain inPDDL+ is used to
construct an explicit hybrid automaton model. Figure 6 de-
picts a fragment of the hybrid automaton model of the rover
recharging domain. In this example the variablecharge
represents the charge level for the rover andrecharge-rate
represents the rate of recharging. The variabled

dtcharge
represents the rate of change ofchargeandcharge′ repre-
sents the value ofcharge after a discrete change. In control
mode:

[at(rover1, A), in−sun(rover1),
d

dt
charge(rover1) = 0]

the rover charge is unchanging (although the rover might
not be fully charged). Theactivate-chargercontrol switch
causes the rover to enter the new control mode in which the
rover is charging and the level of charge is changing ac-
cording to the rate of recharging over time. This mode can

dt
d

dt
d

dt
d

in−sun(rover1)
at(rover1,A) at(rover1,A)

in−sun(rover1)

at(rover1,A)
in−dark(rover1)

sunset

time = sundown

sunrise

(Event)

(Event)

(Event)

charging(rover1)

charge(rover1) = 0 charge(rover1) = recharge−rate(rover1)

charge(rover1) = 0

Figure 6: A fragment of the Rover Domain modelled as a
hybrid automaton

only be maintained while the charge level is less than the
capacity of the rover (this constraint is an invariant condi-
tion) and while the rover is in the sun. If the sun goes down
or the rover is fully charged then charging will stop. Fig-
ure 7 depicts the state transition semantics of the cooling
and heating example shown in figure 3.

Related Work
Representation of, and reasoning with, statements about
time and the temporal extent of propositions has long
been an subject of research in AI including planning re-
search (Allen 1984; McDermott 1982; Sandewall 1994;
Kowalski & Sergot 1986; Laborie & Ghallab 1995; Muscet-
tola 1994; Bacchus & Kabanza 2000). Important issues
raised during the extension ofPDDL to handle temporal
features have, of course, already been examined by other
researchers, for example in Shanahan’s work on contin-
uous change within the event calculus (Shanahan 1990),
in Shoham’s (Shoham 1985) and Reichgelt’s (Reichgelt
1989) work on temporal reasoning and work on non-reified
temporal systems (Bacchus, Tenenberg, & Koomen 1991).
Vila (Vila 1994) provides an excellent survey of work in
temporal reasoning in AI. In this section we briefly re-
view some of the central issues that have been addressed,
and their treatment in the literature, and setPDDL2.1 and
PDDL+ in the context of research in temporal logics.

Several researchers in temporal logics have considered
the problems of reasoning about concurrency, continuous
change and temporal extent. These works have focussed
on the problem of reasoning about change when the world
is described using arbitrary logical formulae. The need to
handle complex logical formulae makes the frame problem
difficult to resolve, and an approach based on circumscrip-
tion (McCarthy 1980) and default reasoning (Reiter 1980)
is typical. TheSTRIPSassumption provides a simple solu-
tion to the frame problem when states are described using
atomic formulae. The classical planning assumption is that
states can be described atomically but this is not a general

Figure 7: A fragment of the Satellite Domain modelled as a hybrid automaton. Active processes are denoted by loops on
the states, corresponding to the processes that contribute non-zero components to the values of the derivatives. Other actions
might cause further energy draining processes to affect the derivative of the energy level.

view of the modelling of change. Although simplifying,
this assumption is surprisingly expressive. The bench mark
domains introduced in the third international planning com-
petition suggest that atomic modelling is powerful enough
to capture some complex domains which closely approxi-
mate real problems. The temporal reasoning issues we con-
front are not simplified as a consequence of having made a
simplifying assumption about how states are updated. We
remain concerned with the major issues of temporal reason-
ing: concurrency, continuous change and temporal extent.

In the development ofPDDL+ we made a basic deci-
sion to consider actions and events as instantaneous state
transitions. This allows us to concentrate on the truth of
propositions at points instead of over intervals. The deci-
sion to consider actions and events as instantaneous state
transitions is similar to that made by many temporal reason-
ing researchers (Shanahan 1990; McCarthy & Hayes 1969;
McDermott 1982). In the context ofPDDL+ the approach
has the advantage of smoothly integrating with the classical
planning view of actions as state transitions.

In the remainder of this section we compare thePDDL

extensions that we propose with previous work in temporal
reasoning by considering the three central issues identified
above. Our objective is not to claim that our extensions im-
prove on previous work, but instead to demonstrate that the
implementation of solutions to these three problems within
the PDDL framework makes their exploitation directly ac-
cessible to planning in a way that they are not when embed-
ded within a logic and accompanying proof theory.

Continuous change
Several temporal reasoning frameworks began with consid-
eration of discrete change and, later, were extended to han-
dle continuous change. For example, in (Shanahan 1990)
Shanahan extended the event calculus of Kowalski and Ser-
got (Kowalski & Sergot 1986) to enable the modelling of
continuous change. This process of extension mirrors the
situation faced in extendingPDDL, where a system mod-
elling discrete change already existed. It is, therefore, inter-
esting to compare the use ofPDDL+ with the use of systems
such as the extended event calculus.

In the sink-filling example that Shanahan de-
scribes (Shanahan 1990) he discusses the issues of
termination of events (self-termination and termination by
other events), identification of the level of water in the
sink during the filling process and the effect on the rate
of change in the level of water in a sink when it is being
filled from two sources simultaneously. The behaviour of
the filling process and its effects on the state of the sink
over time are modelled as axioms which would allow an
inference engine to predict the state of the sink at points
during the execution of the process.

PDDL+ allows the representation of the complex interac-
tions that arise when a sink is filled from multiple indepen-
dently controlled water sources by means of decomposition
into the initiation of filling, the process of filling and its ter-
mination (either by water sources all being turned off, or
by flooding). This model is robust, since it easily accom-
modates multiple water sources, simply modifying the rate
of flow appropriately, which then correctly affects the pro-
cess of filling. In contrast to Shanahan’s extension to the



event calculus, this approach does not require that the fill-
ing process be (at least from the point of view of the logical
axiomatisation) terminated and restarted at a new rate when
a water source is opened or closed, since the process simply
remains active throughout — the change in rate of filling is
reflected in a piecewise-linear profile for the depth of water
in the sink, just as it is in Shanahan’s model.

One of the important consequences of continuous be-
haviour is the triggering of events. In Shanahan’s exten-
sions this is achieved through the axiomatisation of causal
relationships — events are not distinguished syntactically
from actions, but only by the fact that their happening is ax-
iomatically the consequence of certain conditions. We be-
lieve that it is important for a planner to have direct access
to the distinction because it determines what the planner
can do and what consequences it can expect from interac-
tions within the world. Although Shanahan might add addi-
tional axioms to capture the difference, the action-oriented
representation of thePDDL tradition makes the distinction
natural and convenient.

Concurrency

The opportunity for concurrent activities complicates sev-
eral aspects of temporal reasoning. Firstly, it is necessary to
account for which actions can be concurrent and secondly
it is necessary to describe how concurrent activities interact
in their effects on the world.

In most formalisms the first of these points is achieved by
relying on the underlying logic to deliver an inconsistency
when an attempt is made to apply two incompatible actions
simultaneously. For example, the axioms of the event calcu-
lus will yield the simultaneous truth and falsity of a fluent if
incompatible actions are applied simultaneously and conse-
quently yield an inconsistency. Unfortunately, recognising
inconsistency is, in general, undecidable, for a sufficiently
expressive language. InPDDL+ we adopt a solution that ex-
ploits the restricted form of the action-centred formalism,
defining the circumstances in which two actions could lead
to inconsistency (are mutex) and rejecting the simultaneous
application of such actions.

Shanahan (Shanahan 1999) discusses Gelfond’s (Gel-
fond, Lifschitz, & Rabinov 1991) example of the soup bowl
in which the problem concerns raising a soup bowl without
spilling the soup. Two actions, lift left and lift right, can
be applied to the bowl. If either is applied on its own the
soup will spill, but, it is argued, if they are applied simulta-
neously then the bowl is raised from the table and no soup
spills. To model this situation Shanahan uses an explicit as-
sertion of the interaction between the lift left and lift right
actions to ensure that the spillage effect is cancelled when
the pair is executed simultaneously. The assumption is that
the reasoner can rely on the successful simultaneity in order
to exploit the effect.

In PDDL+ we reject this solution on the grounds that pre-
cise simultaneity is outside the control of any physical ex-
ecutive. PDDL+ supports the modelling of the soup bowl
situation in the following way. Lift left and lift right both
independently initiate tilting processes which, after a mea-
surable amount of time, will result in spillage of the soup.
Provided that the two lift actions occur within an appropri-
ate tolerance of one another the tilting will be corrected and
the spillage avoided without the need to model cancellation
of effects. We argue that an executive can execute the two
actions to within a fine but non-zero tolerance of one an-
other, and can therefore successfully lift the bowl.

Temporal extent

A common concern in temporal reasoning frameworks, dis-
cussed in detail by Vila and others (Vila 1994; van Bentham
1983), is thedivided instant problem. This is the prob-
lem that is apparent when considering what happens at the
moment of transition from, say, truth to falsity of a propo-
sitional variable. The question that must be addressed is
whether the proposition is true, false, undefined or inconsis-
tently both true and false at the instant of transition. Clearly
the last of these possibilities is undesirable. The solution
we adopt is a combination of the pragmatic and the philo-
sophically principled. The pragmatic element is that we
choose to model actions as instantaneous transitions with
effects beginning at the instant of application. Thus, the
actions mark the end-points of intervals of persistence of
state which are closed on the left and open on the right.
This ensures that the intervals nest together without incon-
sistency and the truth values of propositions are always de-
fined. The same half-open-half-closed solution is adopted
elsewhere (Shanahan 1999).

Plan Generation and Validation
The semantics we have provided forPDDL+ suggests
that model-checking techniques appropriate for hybrid au-
tomata might provide a basis for the development of plan
generation algorithms forPDDL+. We have not yet explored
the plan generation problem further, but instead have fo-
cussed on the question of validation, since it is essential
that plans be verifiable efficiently.

It is necessary to automate the validation of plans pro-
duced for complex domains since it cannot be considered
reliable or even feasible for non-trivial plans to be checked
by human experts. The validation problem is decidable for
PDDL because plans are finite and can be validated simply
by simulation of their execution. The issue is complicated
for PDDL2.1 andPDDL+ because the validity of a plan de-
pends on confirming that the actions are applicable in the
states that result from any events and/or processes triggered
by the initial state or by actions in the plan. Neither events
nor processes will be visible in the finished plan, so it will



no longer be safe to assume that the actions in the plan will
chain together as they do inPDDL plans (actions might rely
on preconditions that are established by active processes or
events). Plan validation must confirm that actions interact
with any active processes and events to perform a success-
ful state transition between the initial and goal states. Since
the ability to validate plans is crucial for the practicality
of the language, we consider some of the restrictions that
might be imposed onPDDL2.1 andPDDL+ to make valida-
tion decidable.

Plan validation is decidable for domains including dis-
cretized and continuous durative actions, but without events
or processes, because all activity is encapsulated with the
durative actions explicitly identified by a plan. This makes
the trace induced by a plan finite and hence checkable. We
therefore observe that the validation problem forPDDL2.1
is decidable even when actions contain duration inequali-
ties. This is because the work in determining how the du-
ration inequalities should be solved has already been com-
pleted in the finished plan so validation of the plan can pro-
ceed by simulation of its execution, as is the case forPDDL

plans.
The situation is more complex forPDDL+ however. Our

hybrid automaton-based semantics is very powerful and al-
lows us to give meaning to far more plans than we can
validate. In general, given a sequence of actions,P , the
question of whetherP is a valid plan is undecidable. Once
events are introduced, it is possible to create domains in
which entering a certain state can trigger a cascade of events
so that the number of important happenings in the plan is
no longer finite (even though the number of actions still is).
Checking whether such a cascade terminates before the next
action is executed can be undecidable. One solution would
be to allow only a finite number of events to occur between
any two time points (by insisting that they be separated by
a minimum of some fixed amount of time). This results
in a finite trace, and hence a decidable validation problem.
However, this would place an arbitrary constraint on what
domains can be correctly modelled.

Identifying the restrictions under which validation can be
done efficiently is is a focus of our current research.

Conclusions

PDDL+ enables the representation of a class of deterministic
mixed discrete-continuous domains as planning domains.
This is important for the modelling of many realistic prob-
lem domains in which interacting processes arise resulting
in continuous change that cannot be terminated under the
direct control of the planner. We have given some examples
of situations in which this arises and in which simplification
to discrete models would result in failure to capture the true
meaning of the domain. The semantics of the language is
given by means of a mapping showing how hybrid automata

can be constructed fromPDDL+ domains. The power of hy-
brid automata means that we can interpret more plans than
we can efficiently validate and we are currently working on
identifying the conditions under which efficient validation
is possible.

The development of thePDDL sequence towards greater
expressive power is important to the planning community
because thePDDL family of languages has provided a com-
mon foundation for much of the research effort over the past
decade. The problems involved in modelling the behaviour
of mixed discrete-continuous systems have been well ex-
plored but there have been no widely adopted models within
the planning community.PDDL+ begins to bridge the gap
between basic research and applications-oriented planning
by providing the expressive power necessary to capture real
problems.

